AD-A130 079 STORMER EQUATION IMPLEMENTATION USING OFFSET DIPOLE COORDINATES DERIVED F..(U) UTAH UNIV SALT LAKE CITY D J COOKE MAR 83 SCIENTIFIC-3 AFGL-TR-83-0102 1/1

UNCLASSIFIED F19628-81-K-0020 F/G 4/1 NL

END
DATE FILMED
8 83
DTIC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS 1963 A

ADA130079

AFGL-TR-83-0102

STORMER EQUATION IMPLEMENTATION, USING OFFSET DIPOLE COORDINATES DERIVED FROM SPECIFIED GEOGRAPHIC COORDINATES

David J. Cooke

The University of Utah
Salt Lake City, Utah 84112

Scientific Report No. 3

March 1983

Approved for public release; distribution unlimited

AIR FORCE GEOPHYSICS LABORATORY
AIR FORCE SYSTEMS COMMAND
UNITED STATES AIR FORCE
HANSCOM AFB, MASSACHUSETTS 01731

This report has been reviewed by the ESD Public Affairs Office (PA) and is releasable to the National Technical Information Services (NTIS).

This technical report has been reviewed and is approved for publication

Don F. Smart
Contract Manager

WILLIAM J. BURKE, Actg Branch Chf
Plasmas, Paticles & Fields Branch
Space Physics Division

FOR THE COMMANDER

RITA C. SAGALYN, Director
Space Physics Division

Qualified requestors may obtain additional copies from the Defense Technical Information Center. All others should apply to the National Technical Information Service.

If your address has changed, or if you wish to be removed from the mailing list, or if the addressee is no longer employed by your organization, please notify AFGL/DAA, Hanscom AFB, MA 01731. This will assist us in maintaining a current mailing list.

Do not return copies of this report unless contractual obligations or notices on a specific document requires that it be returned.

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER AFGL-TR-83-0102	2. GOVT ACCESSION NO. AD-A130079	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) STORMER EQUATION IMPLEMENTATION, USING OFFSET DIPOLE COORDINATES DERIVED FROM SPECIFIED GEOGRAPHIC COORDINATES		5. TYPE OF REPORT & PERIOD COVERED Scientific Report No. 3 6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) David J. Cooke		8. CONTRACT OR GRANT NUMBER(s) F19628-81-K-0020
9. PERFORMING ORGANIZATION NAME AND ADDRESS The University of Utah Salt Lake City, Utah 84112		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 61102F, Proj 2311, T2311G1 1HWU 2311 G1AP
11. CONTROLLING OFFICE NAME AND ADDRESS Air Force Geophysics Laboratory Hanscom AFB, Massachusetts 01731 Monitor/D.F.Smart/PHG		12. REPORT DATE March 1983 13. NUMBER OF PAGES 13
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified 15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Cosmic rays
Stormer equation
Offset dipole coordinates

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

For many purposes it is useful to have a fast means of estimating directional cosmic ray cutoffs pertaining to any specified direction and location. The Stormer equation offers a means of determining directional cutoffs to a precision suitable for use in non-critical applications, at a speed many orders of magnitude faster than the process of determining precise cutoffs by the "real" geomagnetic field trajectory tracing method. Use of offset dipole coordinates in the Stormer equation allows minimization of

errors in the cutoff estimates arising from the non-inclusion of other than dipolar field terms.

A self contained procedure has been developed for computer use which, for any given location and direction expressed in geographic coordinates, first allows determination of the location and direction relative to the earth's equivalent magnetic dipole (whose location is determined from the lowest order harmonic field terms). The Stormer equation is then invoked, using these offset dipole coordinates, and a cutoff estimate so produced.

This report contains a printed copy of a FORTRAN computer program developed to carry out this task.

STORMER EQUATION IMPLEMENTATION, USING OFFSET DIPOLE
COORDINATES DERIVED FROM SPECIFIED GEOGRAPHIC COORDINATES

David J. Cooke
Department of Physics
University of Utah
Salt Lake City, UT 84112

Introduction: For many purposes it is useful to have a fast means of estimating directional cosmic ray cutoffs pertaining to any specified direction and location. The derivation of "real" geomagnetic field cutoffs by means of computer calculations (see McCracken et al. 1962; Freon and McCracken, 1962; Shea et al, 1965; Cooke and Humble, 1979; for example) is a slow operation, and expensive in computer time. The Stormer cutoff function (Stormer (1930, 1955)), which expresses the dependence of the Stormer cutoff on location and direction in a dipole approximation to the earth's field, offers a means of determining cutoffs which is many orders of magnitude faster than the trajectory tracing method, and one which is sufficiently precise to produce useful estimates of cutoffs in non-critical applications.

The fundamental imprecision in the Stormer expression in representing the real field cutoffs, in particular due to failure to take into account higher order field harmonic terms, or the width and transparency of the penumbra, is normally exacerbated by the use of earth centered geographic or geomagnetic coordinates when invoking the expression (because the earth's equivalent dipole is not earth centered). It is possible to appreciably improve the accuracy of the cutoff estimates by using "magnetic" coordinates (i.e. offset dipole latitude, longitude, zenith, and azimuth) when employing

the expression, and in this way to take into account the offset and tilt of the earth's equivalent dipole. By this means, inherently, the effect of ignoring the higher order field terms is minimized. Smart and Shea (1977) have discussed the advantages of using offset dipole coordinates in conjunction with the Stormer expression, and the use of this expression for interpolating cosmic ray cutoffs over intervals within which precise calculated values do not exist.

A self contained system for transforming from geographic coordinates to offset dipole coordinates is described in this report. A printed copy of the computer program containing this procedure, which uses the calculated coordinates to derive cutoff estimates from the Stormer expression, is appended to the report.

Discussion: The coordinate transformation has a number of stages, which are individually described in the following:

1) A coordinate conversion is used to determine the offset dipole latitude, longitude, and radius from the nominated geographic latitude and longitude, and geocentric altitude. This conversion takes into account the offset and inclination of the earth's equivalent dipole for any required epoch, and assumes that the earth is an oblate spheroid of eccentricity 0.00674. The angle conversion equations are as follows:

$$\text{offset dipole longitude } \phi = \arctan\left(\frac{R \sin \psi \cos \lambda - y \cos b}{C}\right)$$

$$\text{offset dipole latitude } \theta = \arctan(\cos \phi \tan \alpha) \qquad \text{I}$$

$$\text{offset dipole radius } R' = \frac{C}{\cos \theta \cos \phi}$$

where $\alpha = \arctan(F/G) + a$

$$C = \frac{G \cos\alpha}{\cos(\alpha - a)}$$

and $F = R \sin \lambda - x$

$$G = y \sin b + R \cos \lambda \cos (\psi - c)$$

λ and ψ are the geographic latitude and longitude respectively; R is the geographic radius at the specified location; x is the displacement of the equivalent earth dipole above the equatorial plane; y is the displacement of the dipole from the center of the earth in a direction parallel with the geographic equatorial plane; a is the inclination of the dipole axis from a direction parallel to the geographic N-S axis; b is related to the angle between the zero magnetic longitude and the geographic longitudinal direction towards which the dipole is displaced; and c is the geographic longitudinal direction parallel to which the direction zero offset dipole longitude lies.

A procedure for establishing the position of the equivalent dipole with respect to the geocentric coordinate system at any epoch is presented by Smart and Shea (1977). This procedure, which makes use of the low order terms in the spherical harmonic representation of the geomagnetic field, is discussed more fully by Roederer (1972), and Akasofu and Chapman (1975). The values of x, y, a, b and c used in the presently reported coordinate transformation can thus be determined as required. The equivalent dipole position determination is included as a subroutine in the computer program appended to this report.

2) The following steps rely on the use of a particular means of specifying the relative position of two points on the surface of a sphere. In particular, two angles are used, one (σ) defines the angle between the great circle connecting the two points and the meridian line intersecting one of them, and the other (γ) is the angle subtended at the center of the sphere by the two points (see figure 1).

Figure 1. Diagram defining the angles σ and γ used to express the relative position of the two points P_1 and P_2 on a spherical surface. C is the centre of the sphere.



If the two points are, as specified by latitude and longitude, (λ_1,ψ_1) and (λ_2,ψ_2), then σ and γ are given by

$$\sigma = \arctan \left(\frac{\sin(\psi_2 - \psi_1)\cos\lambda_2}{\sin\lambda_1 \cos\lambda_2 - \sin\lambda_1 \cos\lambda_2 \cos(\psi_2 - \psi_1)} \right)$$

II

$$\gamma = \arccos(\cos(\psi_2 - \psi_1)\cos\lambda_1 \cos\lambda_2 + \sin\lambda_1 \sin\lambda_2)$$

Conversely, if the location of one point is known (λ_1,ψ_1, say), then it is possible to determine the latitude and longitude of a second whose position is defined in terms of a specified σ and γ, by means of

$$\text{latitude} = \arcsin(\cos\sigma \sin\gamma \cos\lambda_1 + \cos\gamma \sin\lambda_1)$$

III

$$\text{longitude} = \psi_1 + \arccos\left(\frac{\cos\gamma - \sin\lambda_1 \sin \text{latitude}}{\cos\lambda_1 \cos \text{latitude}} \right)$$

3) The calculation of the magnetic zenith and azimuth uses the strategy of setting up a vector of known length (R") pointing in the direction of interest. The position the tail of the vector, being the location of interest, is of course known in both geographic and magnetic coordinates (the latter determined by step 1), and similarly both geographic and offset dipole radius values pertaining to this point are known.

The geographic coordinates of the position of the head of the vector can be determined by using calculated σ and γ values, derived as follows (which values pertain to the projection of the vector onto a spherical surface).

$$\gamma = \arctan (R'' \sin ze/(R + R'' \cos ze))$$
$$\sigma = az \qquad \text{IV}$$

Having evaluated these angles, the geographic latitude and longitude of the vector head can be calculated using the relationships III. The offset dipole latitude and longitude, and the distance R* of the vector head from the dipole are then calculated using relationships I. Now, having the offset dipole coordinates of both ends of the vector, the angles σ and γ relating the two points (in the offset dipole frame of reference) can be calculated by means of the relationships II. The azimuth of the vector in the offset dipole frame of reference is simply the σ value, whilst the zenith angle is given by

$$ze = \arcos (R^* \cos \gamma - R')$$

where R' is the offset dipole radius of the vector tail (i.e. the distance of the vector tail from the offset dipole center).

Conclusion: By using the coordinate transformation procedure described, the position (latitude, longitude, and radius) of the site location relative to the offset dipole, and the zenith and azimuth pertaining to the direction of interest, may be calculated from the specified geographic coordinates and geocentric altitude. At each step during the derivation of these parameters tests are performed to ensure that calculated angle values lie in the correct quadrant, and appropriate corrections are made if they do not. Having thus determined the angles and distance relative to the equivalent dipole centered frame of reference the Stormer equation can be invoked with greatest possible precision.

The appropriately normalized Stormer expression is as follows:

$$\text{cutoff} = \frac{59.4 \cos^4 \theta}{R'^2 \left(1 + \sqrt{1 - \cos^3 \theta \sin az \sin ze}\right)^2}$$

This expression takes in offset dipole latitude (θ), zenith (ze), azimuth (az, measured clockwise from magnetic north), and radial distance from the effective dipole center (R'); and produces cutoff rigidity values, in units of GV.

The appendix contains a printed copy of the computer program which carries out, in the way described, the entire task of producing a Stormer cutoff estimate from the specified input geographic position and direction parameters.

Acknowledgements

This research was carried out under the sponsorship of the U.S. Air Force Geophysics Laboratory (under contract # F 19628-81-K-0020).

References

Akasofu, S., and S. Chapman, Solar-Terrestrial Physics (Chapter 2), Oxford University Press, London, 1972.

Cooke, D.J., and J. E. Humble, J. Geophys. Res., 75, 5961, 1970.

Freon, A., and K. G. McGracken, J. Geophys. Res., 67, 888, 1962.

McGracken, K. G., U. R. Rao, and M. A. Shea, M.I.T. Technical Report 77, 1962.

Roederer, J. G., Rev. Geophys. and Space Phys., 10. 599, 1972.

Shea, M. A., D. F. Smart, and K. G. McGracken, J. Geophys. Res., 70, 4117, 1965.

Smart, D. F., and M. A. Shea, 15th International Cosmic Ray Conference, Conference Papers, 11, 256, 1977.

Stormer, C., Astrophys., 1, 237, 1930.

Stormer, C., The Polar Aurora, Oxford University Press, London, 1955.

```
      PROGRAM STORM
      DOUBLE PRECISION ZE, AZ, LAT, LON, ZER, AZR, LATR, LONR, ZERO, DTOR
      DOUBLE PRECISION MLATA, MLONA, ZEMA, AZMA, ZENEW, AZNEW, GAM, SIG
      DOUBLE PRECISION DSIN, DCOS, DTAN, DASIN, DACOS, DATAN, DABS, DSQRT
      DOUBLE PRECISION DFLOAT, INCR, ERAD, DIFF, A, B, X, Y, R, RDEL, ALT
      DOUBLE PRECISION GAMVAL, MAGLAT, MAGLON, RMAG, CUTOFF, PI, PID2, PIB2
      COMMON /PIANG/PI, PID2, PIB2, ZERO
      COMMON /ANGLES/ SIG, GAM
      COMMON /POS/ X, Y, A, B, DIFF, ERAD
      COMMON /MAGIO/ ZENEW, AZNEW, R, RMAG, CUTOFF, MAGLAT, MAGLON, GAMVAL,
     X      RDEL, ALT
C ANGLE CONVERSION FACTORS AND CONSTANTS ARE PRESET IN THE FOLLOWING.
      PI = 3.14159265359D+00
      PID2 = PI/2.0D+00
      PIB2 = PI*2.0D+00
      DTOR = PI/180.0D+00
      RDEL = 100.0D+00
      INCR = 1.0D-10
      ZERO = 0.0D+00
C POSITION AND INCLINATION OF OFFSET DIPOLE NOW DEDUCED.
      CALL POSITN
C IF DESIRED, INSERT OTHER VALUES OF A, B, X, Y, DIFF AND ERAD
      A = A*DTOR
      B = B*DTOR
      DIFF = DIFF*DTOR
C INPUT DATA ACCEPTED HERE.
  100 TYPE 1000
      ACCEPT *, LAT, LON, ZE, AZ, ALT
C TERMINATE RUN IF LATITUDE VALUE ENTERED IS OUTSIDE THE POSSIBLE RANGE.
      IF (DABS(LAT) .GT. 90.0D+00) GO TO 200
      ZER = ZE*DTOR+INCR
      AZR = AZ*DTOR+INCR
      CALL RANGE(AZR)
      LATR = LAT*DTOR+INCR
      LONR = LON*DTOR+INCR
C NOW DEDUCE OFFSET DIPOLE COORDINATES OF DIRECTION.
      CALL MAGCO(LATR, LONR, ZER, AZR)
C CALCULATED ANGLES ARE NOW CONVERTED FROM RADIANS TO DEGREES.
      MLATA = MAGLAT/DTOR
      MLONA = MAGLON/DTOR
      ZEMA = ZENEW/DTOR
      AZMA = AZNEW/DTOR
C RESULTS ARE PRINTED OUT BY THE FOLLOWING INSTRUCTIONS.
      WRITE(6,2000)
      WRITE(6,4000)
      WRITE(6,5000) LAT, LON, ZE, AZ, ALT
      WRITE(6,6000) MLATA, MLONA, ZEMA, AZMA, RMAG, CUTOFF
      WRITE(6,3000)
      GO TO 100
 1000 FORMAT(' ENTER LAT, LON, ZE, AZ, AND ALT:  ',$)
 2000 FORMAT(1X)
 3000 FORMAT(1X,/)
 4000 FORMAT (17X,45H LATITUDE   LONGITUDE      ZENITH     AZIMUTH,
     X     36H    ALTITUDE     DIP.RAD.       CUTOFF)
 5000 FORMAT (13H GEOGRAPHIC:  ,5F12.2)
 6000 FORMAT (13H OFFSET DIP.: ,4F12.2,12X,2F12.2)
  200 STOP
      END
```

```
      SUBROUTINE RANGE(PARAM)
C
C THIS SUBROUTINE BRINGS ANGLE VALUES TO WITHIN 0-360 DEGREE RANGE
C
      COMMON /PIANG/PI,PID2,PIB2,ZERO
      DOUBLE PRECISION PARAM,PI,PID2,PIB2,ZERO
      IF (PARAM .GE. PIB2) PARAM = PARAM-PIB2
      IF (PARAM .LT. ZERO) PARAM = PARAM+PIB2
      RETURN
      END



      SUBROUTINE CHECK(PARAM)
C
C THIS SUBROUTINE PREVENTS EFFECT OF ROUND-OFF ERRORS ETC. CAUSING
C THE ARGUMENT OF AN ANGLE FUNCTION TO EXCEED 1.00 MAGNITUDE
C
      DOUBLE PRECISION PARAM
      IF (PARAM .GT. +1.0D+00) PARAM = 1.0D+00
      IF (PARAM .LT. -1.0D+00) PARAM = -1.0D+00
      RETURN
      END



      SUBROUTINE CONV(LAT1,LAT2,LON1,LON2)
C
C THIS SUBROUTINE CALCULATES THE GAMMA AND SIGMA ANGLES RELATING THE
C TWO EARTH- OR DIPOLE-CENTERED VECTORS DIRECTED TOWARDS (LAT1,LON1)
C AND (LAT2,LON2).
C
      DOUBLE PRECISION CHVAL,GAM,SIG,DSIN,DASIN,DCOS,DACOS,PI,PID2,PIB2
      DOUBLE PRECISION LAT1,LAT2,LON1,LON2,VAL,DTLR,GAMD,SIGD,VALD,ZERO
      COMMON /PIANG/ PI,PID2,PIB2,ZERO
      COMMON /ANGLES/ SIG,GAM
      CHVAL = DCOS(LON1-LON2)*DCOS(LAT1)*DCOS(LAT2)+DSIN(LAT1)*
     X      DSIN(LAT2)
      CALL CHECK(CHVAL)
      GAM = DACOS(CHVAL)
      SIG = ZERO
      IF (GAM .EQ. ZERO) GO TO 100
      CHVAL = DSIN(LON2-LON1)*DCOS(LAT2)/DSIN(GAM)
      CALL CHECK(CHVAL)
      SIG = DASIN(CHVAL)
  100 CONTINUE
      CHVAL = DCOS(GAM)*DSIN(LAT1)
      CALL CHECK(CHVAL)
      VAL = DACOS(CHVAL)
      IF ((PID2-LAT2) .GE. VAL) SIG = PI-SIG
      CALL RANGE(SIG)
      RETURN
      END
```

```
      SUBROUTINE CALC(LAT,LON,GAM,SIG,LATT,LONN)
C
C THIS SUBROUTINE DEDUCES THE LATITUDE AND LONGITUDE OF A DIRECTION
C WHICH HAS A NOMINATED SIGMA AND GAMMA DEVIATION FROM A SPECIFIED
C LATITUDE AND LONGITUDE
C
      DOUBLE PRECISION SIG,GAM,LAT,LON,LATT,LONN,FUNC,PI,PID2,PIB2,ZERO
      DOUBLE PRECISION DSIN,DCOS,DASIN,DACOS,DABS,DFLOAT,VAL,CHVAL
      COMMON /PIANG/ PI,PID2,PIB2,ZERO
      CHVAL = DCOS(SIG)*DSIN(GAM)*DCOS(LAT)+DCOS(GAM)*DSIN(LAT)
      CALL CHECK(CHVAL)
      LATT = DASIN(CHVAL)
      FUNC = 1.0D+00
      IF (SIG .GT. PI) FUNC = -1.0D+00
      CHVAL = (DCOS(GAM)-DSIN(LAT)*DSIN(LATT))/DCOS(LAT)/DCOS(LATT)
      CALL CHECK(CHVAL)
      LONN = LON+FUNC*DACOS(CHVAL)
      IF (LATT .LT. PID2) GO TO 100
      LATT = PI-LATT
      LONN = LONN-PI
  100 IF (LATT .GT. -PID2) GO TO 200
      LATT = -LATT-PI
      LONN = LONN-PI
  200 CALL RANGE(LONN)
      RETURN
      END



      SUBROUTINE GETOMA(LAT,LON,RVEC)
C
C THIS SUBROUTINE TAKES IN GEOGRAPHIC LATITUDE & LONGITUDE AND
C OUTPUTS OFFSET DIPOLE LATITUDE & LONGITUDE, LOCAL GEOGRAPHIC
C RADIUS, AND DISTANCE FROM THE OFFSET DIPOLE.
C
      DOUBLE PRECISION PI,PID2,PIB2,A,B,X,Y,DIFF,LAT,LON,ZENEW,AZNEW,ALT
      DOUBLE PRECISION R,RMAG,CUTOFF,MAGLAT,MAGLON,GAMVAL,RVEC,RDEL,VAL
      DOUBLE PRECISION DSIN,DCOS,DTAN,DASIN,DACOS,DATAN,DSQRT,DABS,ZERO
      DOUBLE PRECISION ERAD
      COMMON /PIANG/ PI,PID2,PIB2,ZERO
      COMMON /POS/ X,Y,A,B,DIFF,ERAD
      COMMON /MAGIO/ ZENEW,AZNEW,R,RMAG,CUTOFF,MAGLAT,MAGLON,GAMVAL,
     X       RDEL,ALT
      R = ALT+6356.7747D+00/DSQRT(1.0D+00-0.00673966D+00*
     X       DSIN(PID2-LAT)**2)
      IF (RVEC .GT. ZERO) R = RVEC
      AVAL = R*DSIN(LAT)-X
      VAL = Y*DSIN(B)+R*DCOS(LAT)*DCOS(LON-DIFF)
      GAMVAL = DATAN(AVAL/VAL)+A
      BVAL = VAL*DCOS(GAMVAL)/DCOS(GAMVAL-A)
      MAGLON = DATAN((R*DSIN(LON-DIFF)*DCOS(LAT)-Y*DCOS(B))/BVAL)
      IF (BVAL .LT. ZERO) MAGLON = MAGLON+PI
      CALL RANGE(MAGLON)
      MAGLAT = DATAN(DCOS(MAGLON)*DTAN(GAMVAL))
      RMAG = DABS(BVAL/DCOS(MAGLAT)/DCOS(MAGLON))
      RETURN
      END
```

```
      SUBROUTINE MAGCO(LAT,LON,ZE,AZ)
C
C THIS SUBROUTINE TAKES A GIVEN SET OF GEOGRAPHIC COORDINATES
C (LAT,LON,ZE,AZ) AND DEDUCES THE CORRESPONDING SET OF OFFSET
C DIPOLE COORDINATES
C
      DOUBLE PRECISION A,B,X,Y,ZENEW,AZNEW,R,RMAG,CUTOFF
      DOUBLE PRECISION ZEM,AZM,MAGLAT,MAGLON,GAMVAL,RDEL,RV,CHVAL
      DOUBLE PRECISION DSIN,DCOS,DTAN,DASIN,DACOS,DATAN,LAT,LON,ZE,AZ
      DOUBLE PRECISION MLAT,MLON,EPS,RVEC,GAM,SIG,LATVEC,LONVEC,ALT
      COMMON /POS/ X,Y,A,B,DIFF,ERAD
      COMMON /MAGIO/ ZENEW,AZNEW,R,RMAG,CUTOFF,
     X      MAGLAT,MAGLON,GAMVAL,RDEL,ALT
      COMMON /ANGLES/ SIG,GAM
C DEDUCE OFFSET DIPOLE LATITUDE AND LONGITUDE FROM GEOGRAPHIC VALUES
      CALL SETOMA(LAT,LON,-1.0D+00)
      RV = RMAG
      MLAT = MAGLAT
      MLON = MAGLON
C DEDUCE GAMMA AND SIGMA APPLYING TO SAMPLE VECTOR.
      EPS = DATAN(RDEL*DSIN(ZE)/(R+RDEL*DCOS(ZE)))
      RVEC = (R+RDEL*DCOS(ZE))/DCOS(EPS)
      GAM = EPS
      SIG = AZ
C CALCULATE GEOGRAPHIC LAT. & LON. OF VECTOR HEAD.
      CALL CALC(LAT,LON,GAM,SIG,LATVEC,LONVEC)
C CONVERT THESE GEOGRAPHIC COORDINATES INTO OFFSET DIPOLE COORDINATES.
      CALL SETOMA(LATVEC,LONVEC,RVEC)
C CALCULATE GAM & SIG APPLYING TO OFFSET DIPOLE COORDINATE SYSTEM
      CALL CONV(MLAT,MAGLAT,MLON,MAGLON)
C SO, DEDUCE OFFSET DIPOLE ZENITH AND AZIMUTH .
      AZNEW = SIG
      CHVAL = (RMAG*DCOS(GAM)-RV)/RDEL
      CALL CHECK(CHVAL)
      ZENEW = DACOS(CHVAL)
C NOW COMPUTE CUTOFF VALUE FROM STORMER EXPRESSION.
      CALL CUTCAL(MLAT,MLON,ZENEW,AZNEW,RV,CUTOFF)
      MAGLAT = MLAT
      MAGLON = MLON
      RMAG = RV
      RETURN
      END



      SUBROUTINE CUTCAL(LAT,LON,ZE,AZ,R,CUTOFF)
C
C THIS SUBROUTINE CALCULATES THE STORMER CUTOFF CORRESPONDING TO THE
C SPECIFIED OFFSET DIPOLE LATITUDE, LONGITUDE,ZENITH AND AZIMUTH.
C
      DOUBLE PRECISION LAT,LON,ZE,AZ,R,CUTOFF
      DOUBLE PRECISION DSIN,DCOS,DTAN,DASIN,DACOS,DATAN,DSQRT
      CUTOFF = 59.4D+00*(DCOS(LAT))**4/((R/6400.0D+00)**2*
     X      (1.0D+00+DSQRT(1.0D+00-(DCOS(LAT))**3*DSIN(AZ)*
     X      DSIN(ZE)))**2)
      RETURN
      END
```

```
      SUBROUTINE POSITN
C
C THIS SUBROUTINE DETERMINES THE POSITION OF THE EQUIVALENT DIPOLE
C IN GEOGRAPHIC COORDINATES, AND RETURNS THE DEFINING PARAMETERS
C X, Y, A, B, & DIFF.
C
      DOUBLE PRECISION X,Y,A,B,DIFF,ERAD,PI,PID2,PIB2,ZERO
      DOUBLE PRECISION G01,G11,G02,G12,G22,H01,H11,H02,H12,H22
      DOUBLE PRECISION C11,PHI,PHID,H,L0,L1,L2,E,DENOM,XED,YED,ZED
      DOUBLE PRECISION DSQRT,DATAN,DFLOAT,CUBERT,THETA,THETAD,PSI,PSID
      COMMON /POS/ X,Y,A,B,DIFF,ERAD
      COMMON /PIANG/ PI,PID2,PIB2,ZERO
C FIELD COEFFICIENTS FOR THE EPOCH OF INTEREST ARE INSERTED HERE.
C THE FOLLOWING COEFFICIENTS ARE FOR THE IGRF65 FIELD
      ERAD = 6371.2D+00
      G01 = -0.30339D+00
      G11 = -0.02123D+00
      G02 = -0.01654D+00
      G12 = 0.02994D+00
      G22 = 0.01567D+00
      H01 = 0.0D+00
      H11 = 0.05758D+00
      H02 = 0.0D+00
      H12 = -0.02006D+00
      H22 = 0.00130D+00
      C11 = DSQRT(G11**2+H11**2)
      PHI = DATAN(H11/G11)
      THETA = DATAN(C11/G01)
      H = DSQRT(G01**2+G11**2+H11**2)
      CUBERT = 3.0D+00**(1.0D+00/DFLOAT(3))
      L0 = 2.0D+00*G01*G02+CUBERT*(G11*G12+H11*H12)
      L1 = -G11*G02+CUBERT*(G01*G12+G11*G22+H11*H22)
      L2 = -H11*H02+CUBERT*(G01*H12-H11*G22+G11*H22)
      E = (L0*G01+L1*G11+L2*H11)/(4.0D+00*H*H)
      DENOM = 3.0D+00*H*H
      XED = (L1-G11*E)*ERAD/DENOM
      YED = (L2-H11*E)*ERAD/DENOM
      ZED = (L0-G01*E)*ERAD/DENOM
      DTOR = PI/180.0D+00
      PSI = DATAN(YED/XED)
      IF (THETA .GT. ZERO) GO TO 100
      THETA = -THETA
      PSI = PSI+PI
      CALL RANGE(PSI)
  100 A = THETA/DTOR
      DIFF = PHI/DTOR
      PSID = PSI/DTOR
      B = PSI-PHI-PID2
      CALL RANGE(B)
      CALL RANGE(PHI)
      B = B/DTOR
      DIFF = PHI/DTOR
      X = ZED
      Y = DSQRT(XED*XED+YED*YED)
C CONSTANTS DEFINING THE POSITION AND ORIENTATION OF THE EQUIVALENT
C DIPOLE ARE PRINTED HERE
      WRITE(6,1000) X,Y,A,B,DIFF
 1000 FORMAT(9F12.2)
      RETURN
      END
```